
Chapter 1

Introduction

The conventional approach to programming languages is to manipulate individual primitive
values one at a time. For example, a programming language might have concept of integers,
objects, or functions, but they are almost never manipulated in bulk by the language opera-
tions. Collections of primitive values and operations on them are mostly implemented using
the language facilities themselves. A class of exceptions are array processing languages, and
data-parallel languages that have arrays as a primitive data and many of the language primitives
operate point-wise on them. Other class of exceptions are dataflow languages that manipulate
primitive concepts of the language as infinite streams of values. As such, the types in a data-
flow language denote streams of types, literals of the language are constant streams of values,
and primitive operations manipulate input streams point-wise. Dataflow languages can be used
to model electronic circuits and real-time systems for hardware control. This work focuses on
dataflow languages.

Higher-order functions are a powerful method for abstraction in functional languages. In
dataflow languages they appear to cause some issues. Either the language only implements
first order functions or for example, in Lucid Synchrone the programmer has to explicitly state
if the functional parameter is dataflow dependant. The approach taken by Lucid Synchrone
also implies orthogonality of data types and dataflow types. Reactive programming languages
could potentially overcome those issues but, to our knowledge, so far only higher-order reactive
programming languages are implemented as embedded domain-specific languages in Haskell,
and thus require the dataflow types to be specified explicitly.

Semantics for higher-order dataflow languages, and reactive programming languages exist,
but to our knowledge practical non-embedded implementations do not. Earliest instances of
semantics to higher-order dataflow languages were based on the concept of arrows [Hug00].
Uustalu and Vene proposed in [UV05] that arrows are overly general for this purpose, and
demonstrated that semantics to higher-order dataflow languages can be based on a simpler
concept of comonads. The semantics to functional reactive programming languages has still
been evolving around arrows, especially as the category-theoretic understanding of them has
continuously been improving. As of late it has been shown that linear temporal logic types
functional reactive programs [Jef12]. Curry-Howard correspondence between linear temporal
logic and functional reactive programming has been established.

Both the functional reactive languages embedded in Haskell and comonadic dataflow lan-
guage implementations have yet to enjoy particularly efficient implementations. Both suffer
from issues with large and opaque memory consumption, and the comonadic semantics addi-

tionally has problems with time efficiency due to unnecessary repeated computations.
In order to move closer to an efficient implementation of the comonadic semantics we will

develop semantically equivalent abstract machine [HM92]. The abstract machine is closer to
hardware, and its behaviour is operationally much simpler than the behaviour of the denota-
tional semantics. The abstract machine is intuitively a simple term rewriting system.

The reader is expected to be familiar with basics of the non-strict functional programming
language Haskell [M+10], and the basics of the theory of programming language semantics.
Language evaluators, various program transformations and examples are presented in Haskell.
While strictly not required it helps if reader is familiar with the concept of monads, arrows, or
comonads. Familiarity with λ-calculus, continuation-passing style transformation, thunk-based
simulation, and defunctionalization could also be of help but is not required.

1.1 Outline
In Chapter 2 we give an overview of comonadic dataflow languages. We start by presenting a
short introduction to programming in a dataflow language Lucid Synchrone to provide some
context. Next, we will introduce the notion of comonads, and present a comonadic evaluator
for a higher-order non-strict dataflow language.

The Chapter 3 presents the methodology developed by Ager, Biernacki, Danvy, and Midt-
gaard for deriving abstract machines from denotational evaluators in [ABDM03]. We start by
defining the notion of abstract machines based on [HM92] and present two abstract machines
for evaluating λ-calculus terms as examples. The first is the CEK abstract machine for call-by-
value evaluation, and the other is the Krivine’s abstract machine for call-by-name evaluation.
The Danvy’s methodology uses two major code transformations. First is the continuation-pass-
ing style transformation, and the second is the defunctionalization. We shall introduce these,
and the thunk-based simulation in Section 3.2. Finally, we will derive an abstract machine for
an extremely simple expression-language evaluator to demonstrate the methodology.

The Chapter 4 focuses entirely on deriving an abstract machine for the comonadic evaluator
defined in the second chapter. The derivation goes through the following steps: a) higher-order
functions in value domain are eliminated, and become closures (function objects) via defunc-
tionalization; b) the evaluator is simplified and the comonad specific constructs are inlined;
c) call-by-name behaviour is simulated via thunks; d) call-by-value continuation-passing style
transformation is applied to reach a tail-recursive evaluator; and e) higher-order functions are
eliminated by defunctionalization. The call-by-value continuation-passing style transformation
can be applied because thunk-based simulation results in evaluation order indifferent program.
Finally, the abstract machine, which we show is equivalent to the final evaluator, is presented.
Each evaluator in the derivation sequence is equivalent to the previous one, and thus the final
abstract machine is equivalent to the initial denotational evaluator. The equivalence of each
evaluator to the previous one is due to the correctness of the used program transformations.

1.2 Contributions
As a major result we present an abstract machine for a higher-order non-strict dataflow lan-
guage. The abstract machine can be a basis for future efficient implementations, and gives
insight to the operational behaviour of the language programs. The second contribution is that

we demonstrate that the Danvy’s methodology [ABDM03] can be applied to evaluators imple-
mented in a non-strict host language. While Danvy has shown in [Dan04] that the methodology
results in both the CEK, and the Krivine’s machine if applied to call-by-name evaluator imple-
mented in a strict host language. We will propose that the same result is reached from evaluator
implemented in a non-strict language. This result is significant because the implementation of
a call-by-name evaluator is much simpler and corresponds more directly to the denotational
semantics in a call-by-name host language.

The resulting abstract machine for the non-strict dataflow language is similar to the Kriv-
ine’s abstract machine. Compared to the Krivine’s machine our machine has two extra rules for
evaluating the additional dataflow language construct, and uses overloaded notion of environ-
ment updating and environment lookup. The two extra rules are simple and have constant-time
behaviour. While this result is not surprising it does give additional guarantee of correctness.
We do not present operational semantics to the dataflow language, but the similarity of the ab-
stact machine to the Krivine’s machine suggests that the operational semantics are too similar
to the standard call-by-name λ-calculus operational semantics.

Chapter 2

Comonadic dataflow languages

In this chapter we give an overview of the semantics of a comonadic dataflow language. We
start with giving an overview of dataflow languages, comonads, and then we present a denota-
tional comonadic semantics for a simple causal higher-order dataflow language. The semantics
are based on the works by Uustalu and Vene in [UV05].

2.1 Dataflow languages
Where conventional programming languages manipulate individual values, dataflow languages
on the other hand are designed to manipulate infinite streams of data, which can be viewed as
a discrete-time signals. In this section we give short overview of dataflow languages based on
some examples of Lucid Synchrone [Pou06].

Lucid Synchrone is functional, strict, and strongly typed Objective Caml like programming
language that manipulates infinite sequences as primitive values. The language is extended
with many dataflow primitives, and the type system is extended to support the added primitives.
For example, at the type system level the notion of clocks supplies a way to specify different
execution rates in a program. In this section we explore some simpler concepts of the language
in order to gain some intuition in dataflow programming.

Because this section acts as a basic introduction to dataflow languages we will completely
ignore more advanced features of the language. For example, we will gloss over the type system
entirely by only looking at primitive types, and focus on some of the simpler dataflow features
of the language.

2.1.1 Point-wise operations
Every type and every scalar values in Lucid Synchrone is implicitly lifted to streams. For
instance:

• int stands for the type of integer streams;

• language literal expressions stand for constant streams of values;

• pairs of values denote streams of pairs; and

• arithmetic expressions, such as x + y, operate point-wise on input streams.

Program execution can be represented as chronograms, showing the sequence of values
taken by streams during the execution. The chronogram below shows five streams. The first
column denotes the expression, and rest of the columns denote the values that the expression
takes. Below we have defined stream of booleans c, two integer streams x and y, and an if-
expression.

Expression Value
c true false true true false . . .
x x0 x1 x2 x3 x4 . . .
y y0 y1 y2 y3 y4 . . .

x + y x0 + y0 x1 + y1 x2 + y2 x3 + y3 x4 + y4 . . .
if c then x else y x0 y1 x2 x3 y4 . . .

The let-expression is used to define new streams and functions. For example, to define two
constant streams, and a function to average two input streams point-wise we could write:

Listing 2.1: Simple streams
let dt = 0.001
let g = 9.81
let average (x,y) = (x + y) / 2

Function application is denoted with space, and to compute the point-wise average of streams
dt, and g we can write average (dt,g). Blocks of mutually recursive functions can be defined
using let-expression with the rec, and and keywords:

Listing 2.2: Mutual recursion
let rec odd n = if n == 0 then false else even (n - 1)

and even n = if n == 0 then true else odd (n - 1)

So far we have only seen how to define new constant streams, and how to manipulate
streams point-wise. In order to escape the world of constant streams we will introduce some
dataflow language specific constructs.

2.1.2 Delays
Lucid Synchrone is causal dataflow language: values that a function takes may only depend on
either the current, or previous values of input streams. The language supports many operators
to introduce stream delays. Most important for us is the followed-by operator. The expression
x fby y reads as “x followed by y”, and takes first value of the stream x and after that a previous
value of stream y. In another words, it delays y by a single instant, and is initialized by the first
value of x. The following chronogram demonstrates the behaviour of the followed-by construct:

x x0 x1 x2 x3 x4 . . .
y y0 y1 y2 y3 y4 . . .

x fby y x0 y0 y1 y2 y3 . . .

Lucid Synchrone also has a conceptually simpler delay operation pre that delays an input
stream by a single instant. The expression pre x, reads as “previous x”, is undefined at the
first position and at every other position has the value of stream x at the previous position. The
following chronogram shows the effect of the previous-operator:

x x0 x1 x2 x3 x4 . . .
pre x ⊥ x0 x1 x2 x3 . . .

A static analysis is used to guarantee that undefined values introduced by pre are never used
directly.

To eliminate the undefined behaviour introduced by previous-operator an operator is sup-
plied to replace the first element of an input stream with a well defined value:

x x0 x1 x2 x3 x4 . . .
y y0 y1 y2 y3 y4 . . .

x -> y x0 y1 y2 y3 y4 . . .

The followed-by expression x fby y can now be redefined as x -> pre y.
We call an expressions sequential if it may produce time evolving value when the free

variables are kept constant. Otherwise, the expression is call combinatorial. While there are
other methods to construct sequential expressions for our needs the expression is combinatorial
if it does not contain any of the delay operators. In Lucid Synchrone it is required to anno-
tate let-expressions that introduce sequential functions into global scope with node keyword.
Sequential non-functional values may not be bound with a let-expression in the global scope.

2.1.3 Examples

At this point we are equipped with powerful enough tools to define more interesting streams.
One of the simplest examples is the stream of natural numbers. While we can not directly
declare a stream of integers, we overcome this by defining the stream as a sequential function
from the unit type to integers.

To define stream of integers we need to exploit recursion. Stream of natural numbers starts
with value 0, and is followed by stream of natural numbers with 1 larger value. This gives us
rather straightforward implementation:

Listing 2.3: Stream of natural numbers
let node nat () = nat’ where

rec nat’ = 0 fby (nat’ + 1)

Note the use of the where keyword to define local variables.
Another standard example is stream of factorials of natural numbers, which can be defined

as 1 followed by factorials multiplied by the stream of natural numbers that starts with 1:

Listing 2.4: Factorials
let node fact () = fact’ where

rec fact’ = 1 fby (fact’ * (nat () + 1))

As for slightly more complicated example we will define a stream of Fibonacci numbers.
The stream starts with 1, and the value of every other position is the sum of previous two. To
compute the sum of two previous positions we add the stream to 0 followed by the original
stream. The constant 0 acts as an element before the initial value 1.

Listing 2.5: Fibonacci numbers
let node fib () = fib’ where

rec fib’ = 1 fby (fib’ + (0 fby fib’))

To demonstrate how fib is computed we show the intermediate values that its subexpressions
take in the following table:

fib’ 1 1 2 3 5 8 . . .
0 fby fib’ 0 1 1 2 3 5 . . .

fib’ + (0 fby fib’) 1 2 3 5 8 13 . . .
1 fby (fib’ + (0 fby fib’)) 1 1 2 3 5 8 . . .

2.2 Comonads

Monad is a well known abstraction in functional programming community, and as shown by
Moggi [Mog90, Mog89], and Wadler [Wad92] can be used to provide additional (effectful)
structure for denotational semantics. However, its category theoretic dual, comonad, has not
enjoyed nearly as much use. Monads are widely used to represent effectful computations and
input/output in Haskell, but there is no well known practical use for its dual. While various con-
text dependent computations are known to have a comonadic structure the general abstraction
is almost never needed in practise.

In Haskell comonads can be represented as the following type class:

class Comonad w where
counit :: w a→ a
cobind :: (w a→ b)→ w a→ w b

The class declaration states that each instance of comonad has to define function counit, which
is used to retrieve a value from comonadic structure, and cobind, that is used to lift comonadic
functions. Functions of type w a → b, where w is a comonad, are known as co-Kleisli arrows
from type a to type b.

The type class requires programmer to supply properly typed functions for every comonad
instance, but does not express the laws the functions have to abide to. It’s programmers respon-
sibility that for every defined comonad w, and for every function f :: w b→ c, and g :: w a→ b
the following three laws hold:

cobind counit ≡ id
counit ◦ cobind f ≡ f
cobind f ◦ cobind g ≡ cobind (f ◦ cobind g)

Concrete examples of comonads include: identity comonad, product comonad, costate co-
monad, and non-empty list comonad. Surprisingly cellular automaton can be viewed as a co-
monad [Pip06, CU10], and non-empty arrays with one element focused can also be viewed as
such [Pip08]. In fact, in Haskell, every functor gives rise to a comonad via the cofree comonad
construction.

2.2.1 Product comonad
The product data type:

data Prod e a = Prod e a

is a classic example that give rise to a family of comonads that can be easily interpreted as a
value in a context. The implementation is symmetrical, but the first component can be viewed
as the context, and second component as the value.

For every type e the partial application Prod e defines a comonad:

instance Comonad (Prod e) where
counit (Prod x) = x
cobind d (Prod e x) = Prod e (d (Prod e x))

The unit operation extract the value from context, and the binding operation applies the function
to the input product, and inserts the result back into the context supplied by the input.

The proofs for the comonad laws are easy for the first two, and simply account to expanding
the definitions of counit and cobind. For first law we substitute the right-hand side of counit
and then cobind as follows:

cobind counit (Prod e x)
≡ Prod e (counit (Prod e x))
≡ Prod e x

For the second rule we do the same in reverse order:

counit (cobind f (Prod e x))
≡ counit (Prod e (f (Prod e x))
≡ f (Prod e x)

The correctness of the last rule is somewhat more complicated to show. First, we will
expand both cobind operations, and reintroduce the innermost cobind. The second step is to
introduce function composition of f , and cobind g by rewriting the right-hand side of func-
tion composition operator with the left side. Finally, the outermost cobind is reintroduced by
replacing the right-hand side of the definition with the left giving exactly what is required.

cobind f (cobind g (Prod e x))
≡ cobind f (Prod e (g (Prod e x)))
≡ Prod e (f (Prod e (g (Prod e x))))
≡ Prod e (f (cobind g (Prod e x)))
≡ Prod e ((f ◦ cobind g) (Prod e x))
≡ cobind (f ◦ cobind g) (Prod e x)

Note that we can’t skip expanding the innermost cobind g, as to expand the outer cobind f we
need to know that it was passed value constructed with the Prod data constructor.

It’s important to note that this proof is not quite sufficient as we have not accounted for
undefined values. Luckily, the proof is straightforward to extend, and we shall not go through
this.

While strictly not always required due to [DHJG06] we will expect that the data structures
are defined, and the comonad instance implemented, in such manner that the laws hold for
undefined values too. An example of an incorrect definition of the lifting operator for the
product comonad would be:

cobind d p = Prod e (d p)
where Prod e x = p

because cobind counit ⊥ would evaluate to Prod ⊥ ⊥ which, in Haskell, is distinguishable
from undefined values.

Every comonad has to be equipped with operations that distinguish it from others. For
product comonad we can ask for the context, and perform local computations on it:

ask :: Prod e a→ e
ask (Prod e ) = e

local :: (e→ e′)→ Prod e a→ Prod e′ a
local f (Prod e a) = Prod (f e) a

2.2.2 Non-empty list comonad
Non-empty lists are a more complex example of comonads. They can be constructed by either
taking a single value to construct a singleton list, or appending a value to already existing non-
empty list giving list of at least two elements long. This can be represented by the following
data structure:

data LV a = One a | Cons a (LV a)

The data structure either constructs a singleton list tagged with One, or longer list from already
constructed non-empty list with data constructor Cons. This is very close to the definition
of regular lists with the difference that instead of constructing an empty list we construct a
singleton list.

The unit of comonad can be defined because it’s always possible to extract one element
from a non-empty list. The unit extracts the first value:

instance Comonad LV where
counit (One x) = x
counit (Cons x xs) = x

The definition for lifting is more complicated, but it simply constructs a new non-empty list by
applying the argument function to all the tails of the input list. Resulting list is of same length
as the input list.

cobind d (One x) = One (d (One x))
cobind d (Cons x xs) = Cons (d (Cons x xs)) (cobind d xs)

Proofs for the laws are by induction on the lists length, and take some simple equational
reasoning. We only prove the first law, but the rest follow in quite similar vein.

Proposition 1. For non-empty lists cobind counit ≡ id.

Proof. For bottom-valued lists the equation holds due to strictness of cobind. We show that the
claim holds for singleton lists. Let x be an arbitrary value of some type a.

cobind counit (One x)
≡ { definition of cobind }

One (counit (One x))
≡ { definiton of counit }

One x

Assume that the proposition holds for a non-empty list xs :: LV a. Let x be a value of type
a. We will show that the proposition holds for list Cons x xs by the following reasoning:

cobind counit (Cons x xs)
≡ { definition of cobind }

Cons (counit (Cons x xs)) (cobind counit xs)
≡ { definition of counit }

Cons x (cobind counit xs)
≡ { induction assumption }

Cons x xs

�

Every comonad has to be associated with comonad specific operations. For non-empty lists
we define so called followed-by construct:

fby :: a→ LV a→ a
fby x (One y) = x
fby x (Cons y ys) = counit ys

The meaning of the construct might be difficult to extract, but it will become clear if we look
at an example of its use. The expression cobind (fby x), for example, shifts the argument list
to left, and replaces the last element with x. Let us view a sequence of values x1 x2 . . . xn in
reverse as Cons xn (...(Cons x2 (One x1))). Using such view we can represent the behaviour of
the example as follows:

y y0 y1 y2 y3 y4

cobind (fby x) y x y0 y1 y2 y3

We define two additional functions. First is cmap for applying a function f over every
element of the input list. It is defined as cobind (f ◦ counit), but we give it more explicit but
equivalent definition:

cmap :: (a→ b)→ LV a→ LV b
cmap f (One x) = One (f x)
cmap f (Cons x xs) = Cons (f x) (cmap f xs)

The second function is for zipping two lists together, and is defined as follows:

czip :: LV a→ LV b→ LV (a, b)
czip (Cons x xs) (Cons y ys) = Cons (x, y) (czip xs ys)
czip xs ys = One (counit xs, counit ys)

As the zipping function defined in the Haskell prelude the czip for non-empty lists discards
elements from the longer one.

2.3 Comonadic denotational semantics for a dataflow lan-
guage

In this section we present an evaluator for a causal higher-order dataflow language. The eval-
uator is equivalent to that presented in the works by Uustalu and Vene [UV05], but is slightly
simplified. The first simplification is that of the data structure, and the second is elimination of
a type class as we are only dealing with a single evaluator instead of multiple.

The language is based on lambda calculus and has variables, integer literals, lambda-
expressions, function application, and additionally followed-by construction to introduce stream
delays. The language is not given concrete syntax, and is simply represented in Haskell by the
following abstract syntax tree:

data Term
= Lit Int
| Var String
| Lam String Term
| App Term Term
| Fby Term Term

Everything but the followed-by construction is a standard for a lambda-calculus. The language
does not include fixed point combinators as those can be implemented in a non-typed setting
easily.

The value domain of the language is composed of integers, and functions mapping values
to values. Because the semantics of the terms are given in a co-Kleisli category the functions
need to be co-Kleisli arrows of type LV Value → Value. The data constructor LV is one for
non-empty lists defined previously in Subsection 2.2.2. The data type to represent the value
domain of the language is defined in Haskell as follows:

data Value
= I Int
| F (LV Value→ Value)

The environment maps variables to values, and is implemented as a list of pairs composed
of a variable, and a value:

type Env = [(String,Value)]

We define generic functions for looking up variables, and for inserting values into the envi-
ronment. To map a variable to a value the variable-value pair is simply inserted into beginning
of the list. We also provide uncurried version of insert and name it repair:

insert :: k → v→ [(k, v)]→ [(k, v)]
insert x v e = (x, v) : e

repair :: k → (v, [(k, v)])→ [(k, v)]
repair x = uncurry (insert x)

The operator ! is used to look up inserted values from the environment. The function com-
pares the variable to the first components of the pairs stored in the environment until a match is
found. The second component of the first match is returned:

(!) :: Eq k ⇒ [(k, v)]→ k → v
[ ] ! x = error "Type error!"
((k, v) : kvs) ! x | k ≡ x = v

| otherwise = kvs ! x

If the variable is not found in the environment a type error is raised. Those run-time errors can
be eliminated by performing a simple static check to guarantee that all used variables are within
scope. The variable lookup and insertion are implemented such that the same variable can be
inserted multiple times into the same environment, and the latest inserted one will be returned
by the lookup function. This gives us variable shadowing.

The denotation of terms are given by co-Kleisli arrow from environment to value. The
evaluator follows from [UV05], and directly corresponds to the denotational semantics:

eval :: Term→ LV Env→ Value
eval (Lit n) de = I n
eval (Var x) de = counit de ! x
eval (Lam x t) de = F (λdv→ eval t (repair x ‘cmap‘ czip dv de))
eval (App t0 t1) de = unF (eval t0 de) (cobind (eval t1) de)
eval (Fby t0 t1) de = fby (eval t0 de) (cobind (eval t1) de)

The comonadic operations counit, cobind, cmap, czip and fby have been defined previously in
Subsection 2.2.2.

The helper function unF is used to unwrap function from value data type. The operation
may fail if a non-function valued expression is used in application, but all such cases can be
caught by a type checker.

unF :: Value→ (LV Value→ Value)
unF (I n) = error "Type error!"
unF (F f ) = f

To compute a value of a term e at some stream position i we apply the evaluator to the term
e and a length i non-empty constant list of initial environments. The symbol ε is used to denote
the initial environment.

runEval :: Term→ Int → Value
runEval t i = eval t (ε ↑ i)
(↑) :: a→ Int → LV a
x ↑ 1 = One x
x ↑ n = Cons x (x ↑ (n − 1))

2.3.1 Code examples
In order to demonstrate the evaluator we are going to define the initial environment with some
primitive operations predefined. The operations are represented as functional values.

ε :: Env
ε = [("add", binOp (+)), ("mul", binOp (∗))]

The helper function binOp is used to lift arbitrary binary integer functions into value domain.

binOp :: (Int → Int → Int)→ Value
binOp op = F (λdv0 → F (λdv1 → I (unI (counit dv0) ‘op‘ unI (counit dv1))))

where
unI (I x) = x
unI (F f ) = error "Type error!"

While it’s possible to construct some simple programs with given terms, and primitive op-
erations; we need a fixed-point combinator to define recursive programs. Because the language
is not typed it’s possible to define the fixed-point operator using existing terms. The fixed point
term corresponds the lambda calculus Y combinator λ f .(λx. f (xx))(λx. f (xx)), and is defined by
the following term:

Y :: Term
Y = Lam "f" (App

(Lam "x" (App (Var "f") (App (Var "x") (Var "x"))))
(Lam "x" (App (Var "f") (App (Var "x") (Var "x")))))

A simple example of dataflow program that can not be defined without recursion is one to
compute the stream of natural numbers. The term to do just that can be defined as follows:

nat :: Term
nat = App Y (Lam "nat" (Fby (Lit 0)

(App (App (Var "add") (Lit 1)) (Var "nat"))))

To run the program we evaluate expression map (runEval nat) [1 . .], which evaluates to infinite
list of values [I 1, I 2, . .].

The terms presented directly in Haskell using the abstract syntax tree are quite difficult to
read. In the rest of this section we are going to use some syntactic sugar. The previously defined
terms in the new form look as follows:

Y = λ f . (λ x . f (x x)) (λ x . f (x x))
nat = Y (λ nat . Fby 0 (add 1 nat))

The lambda function constructors have been presented in a nicer form. Application construc-
tors, variable constructors, and literal constructors have been removed. It’s always clear from
the context which constructor has been applied. Parentheses have been used to resolve the cases
where function application constructors could be applied ambiguously. The function applica-
tion is left-associative.

Factorial function can be defined by reusing the stream of natural numbers as follows:

fact :: Term
fact = Y (λ fact . Fby 1 (mul fact (add nat 1)))

Finally, to complete the examples presented in the section about data-flow languages here
follows the term to compute stream of Fibonacci numbers:

fib :: Term
fib = Y (λ fib . Fby 1 (add fib (Fby 0 fib)))

Note that the examples presented here correspond exactly to the examples presented in the
Section 2.1 about dataflow languages.

Chapter 3

From evaluator to abstract machine

3.1 Abstract machines
There are many definitions of abstract machines, and countless different abstract machines
following any of these definitions. Well known examples of abstract machines are the CEK
machine [FF86], Krivine’s machine [Kri85], CLS machine [HM92], SECD machine [Lan64],
and STG machine [PJS89]. We follow the definition by Hannan and Miller [HM92] which
states that an abstract machine is a term rewriting system with a strict initialization condition
and halting conditions, and rewrite rules that: a) are deterministic; b) are always applied to a
term at its root; and c) must not contain repeated variables in the left.

We also distinguish abstract machines from virtual machines. The latter operate on an
instruction set while the former directly on a source language terms. Virtual machines have an
additional compilation step before evaluation.

Following our definition a set of mutually tail recursive functions, operating on language
terms, under strict evaluation, is an abstract machine. If the set of mutually recursive functions
is evaluator for some language, and we can prove that initial evaluator is equivalent to said
set of recursive functions then we can say that the evaluator is equivalent to the corresponding
abstract machine.

3.1.1 The CEK abstract machine
First example is the CEK abstract machine which evaluates lambda calculus terms with call-
by-value strategy. The language terms are denotes with t, variables with x, expressible values
with v, evaluation contexts with k, and environments with env. The expressible values are
closures consisting of a variable, a term, and an environment. The evaluation context is either:
a) a stop marker; b) a function marker storing a closure value, and an evaluation context;
or c) an argument marker consisting of a term, an environment, and an evaluation context.
Environment e maps variables to expressible values. Terms, values, and contexts are described
by the following abstract syntax:

t ::= x | λx.t | t0 t1

v ::= [x, t, e]
k ::= stop | fun(v, k) | arg(t, e, k)

The CEK machine has initial transition, final transition, three evaluation rules, and two
continuation rules for reducing the evaluation context:

t ⇒eval 〈t, nil, stop〉
〈x, e, k〉 ⇒eval 〈k, e(x)〉

〈λx.t, e, k〉 ⇒eval 〈k, [x, t, e]〉
〈t0 t1, e, k〉 ⇒eval 〈t0, e, arg(t1, e, k)〉

〈arg(t1, e, k), v〉 ⇒cont 〈t1, e, fun(v, k)〉
〈fun([x, t, e], k), v〉 ⇒cont 〈t, e[x 7→ v], k〉

〈stop, v〉 ⇒cont v

The three evaluation rules operate on triples composed of a term, an environment, and an
evaluation context. The continuation rules operate on pairs composed of an evaluation context,
and a value.

The evaluation rule pattern matches on the term structure, and if the term is a function
application then the evaluation context is updated to evaluate the function parameter. Otherwise
the evaluation rule transitions into reducing the evaluation context with either the closure value
in case of lambda abstraction, or the value of the variable in case of the variable expression.

The continuation rules, however, pattern match on the evaluation context. If the evaluation
context is a function argument, then the expressible value v is known to be a functional value
and the rule transitions into evaluating the argument with the given functional value on the
stack. If the context is a function object, then the value is known to be argument to the function
and the rule transitions into evaluating the body of the function with environment updated to
map the lambda variable to the value.

3.1.2 Krivine’s abstract machine

The second example is the Krivine’s abstract machine for evaluating lambda terms under call-
by-name (CBN) evaluation strategy. Compared to the CEK machine the Krivine’s machine op-
erates on lambda terms that represent variables with De Bruijn indices. Variables are expressed
with natural numbers denoting the number of in scope lambda binders until one corresponding
to the variable. Lambda terms no longer introduce variables by name. Expressible values v are
pairs composed of a term, and an environment. Environment e is a sequence of values. Stack
of values is denoted with s.

t ::= n | λt | t0 t1

v ::= [t, e]
s ::= • | v :: s
e ::= nil | v :: e

The abstract machine has transition for initialization, transition for halting, and three tran-
sition rules for evaluating the expression.

t ⇒ 〈t, nil, •〉
〈n, e, s〉 ⇒ 〈t, e′, s〉 where [t, e′] = en

〈λt, e, [t′, e′] :: s〉 ⇒ 〈t, [t′, e′] :: e, s〉
〈t0 t1, e, s〉 ⇒ 〈t0, e, [t1, e] :: s〉
〈λt, e, •〉 ⇒ [t, e]

In case of variable n a pair consisting of a term and an environment is looked up from the
currently active environment, and the evaluation is continued with the resulting term and en-
vironment. In case of lambda abstraction the parameter is assumed to be stored on the stack,
and the environment is simply updated to map to the parameter. In case of function application
the parameter is pushed onto the stack with currently active environment, and the evaluation is
continued with the function in the unmodified environment.

3.2 Code transformations
The derivation from denotational evaluator to abstract machine uses two non-trivial code trans-
formations. The first transformation is to convert the program into a continuation-passing style
resulting in an equivalent tail-recursive program. The second transformation is called defunc-
tionalization and is used to eliminate higher-order functions from the input program. In this
section we present both of the transformations in mostly informal manner, and demonstrate the
use of them based on simple examples.

3.2.1 CPS transformation

The continuation-passing style (CPS) transformation is a code transformation that converts
functions into a form in which the control is passed explicitly in the form of a continuation.
The transformation is extremely useful as it converts the function into a tail-recursive, but
higher-order, form. This section uses the results by Plotkin from [Plo75], but is mostly based
on the works by Hatcliff and Danvy in [HD97].

The CPS transformed program has a fixed control flow, and is indifferent to the evaluation
strategy. Informally this means that it does not matter if the transformed program is evaluated
under call-by-name, or call-by-value. Depending on which behaviour of the underlying pro-
gram needs to be preserved either call-by-name, or call-by-value CPS transformation can be
applied.

To formally define both of the CPS transformations we consider Λ, the untyped λ-calculus.
The λ-calculus term t is either a literal constant c, a variable x, a lambda-abstraction, or a
function application.

t ∈ Λ

t ::= c | x | λx.t | t0 t1

The set of values is defined depending on which evaluation strategy the terms are under. Under
call-by-name the value v is either a constant, or a lambda abstraction; and under call-by-value
the value v is either a constant, a variable, or a lambda abstraction.

v ∈ Valuesn[Λ] v ∈ Valuesv[Λ]
v ::= c | λx.t v ::= c | x | λx.t

The term evaluation functions evaln, and evalv are defined in terms of reflexive, transitive
closure of small-step operational semantic rules defined in Figure 3.1, and Figure 3.2 respec-
tively.

evaln(t) = v iff t 7−→∗n v
evalv(t) = v iff t 7−→∗v v

(λx.t0) t1 7−→n t0[x := t1]

t0 7−→n t′0
t0 t1 7−→n t′0 t1

Figure 3.1: Call-by-name small-step operational semantics

(λx.t0) v 7−→v t0[x := v]

t0 7−→v t′0
t0 t1 7−→v t′0 t1

t1 7−→v t′1
(λx.t0) t1 7−→v (λx.t0) t′1

Figure 3.2: Call-by-value small-step operational semantics

Note that both of the small-step transition rules, and both of the evaluation functions operate
on closed lambda terms (terms with no free variables).

We say that two lambda terms are equivalent t0 ' t1 if either both t0, and t1 are undefined,
or both are defined and are equal up renaming of the variables. We write t0 'βi t1 if either both
terms are undefined, or are defined and denote βi-convertible terms. By βi-convertibility we
don’t strictly have β-reduction in mind, but arbitrary small-step operational semantic rules. For
example, a pair of terms might be n-convertible to state that they are convertible under call-by-
name small-step rules. Informally two terms are convertible under a reduction strategy if they
will eventually coverge into same terms.

Both of the CPS transformations are defined as functions taking lambda terms to lambda
terms. The call-by-value transformation Cv, and call-by-name transformation Cn are defined in
Figure 3.3.

Plotkin showed in [Plo75] that both call-by-name, and call-by-value CPS transformed closed
terms are indifferent to the evaluation strategy. Given identity function I this is stated formally
as follows:

evalv(Cn〈[t]〉 I) ' evaln(Cn〈[t]〉 I)
evalv(Cv〈[t]〉 I) ' evaln(Cv〈[t]〉 I)

A second property that Plotkin [Plo75] showed was that call-by-name behaviour is simulatable
in call-by-value setting by performing call-by-name CPS transformation. The other way around
call-by-value can be simulated in call-by-name setting.

Cn〈evaln(t)〉 'βi evalv(Cn〈[t]〉 I)
Cv〈evalv(t)〉 ' evaln(Cv〈[t]〉 I)

Example

Informally, the call-by-value CPS transformation consists of naming all significant interme-
diate results of terms, sequentializing their computation, and introducing continuations. For
example, let us consider non-tail-recursive factorial function defined in Haskell:

fact :: Int → Int
fact n = if n ≡ 1 then 1

else n ∗ fact (n − 1)

Cn〈[·]〉 : Λ→ Λ

Cn〈[v]〉 = λk.k Cn〈v〉
Cn〈[x]〉 = λk.x k

Cn〈[t0 t1]〉 = λk.Cn〈[t0]〉(λy0.y0 Cn〈[t1]〉 k)

Cn〈·〉 : Valuesn[Λ]→ Λ

Cn〈c〉 = c
Cn〈λx.t〉 = λx.Cn〈[t]〉

Cv〈[·]〉 : Λ→ Λ

Cv〈[v]〉 = λk.k Cv〈v〉
Cv〈[t0 t1]〉 = λk.Cv〈[t0]〉(λy0.Cv〈[t1]〉(λy1.y0 y1 k))

Cv〈·〉 : Valuesv[Λ]→ Λ

Cv〈c〉 = c
Cv〈x〉 = x

Cv〈λx.t〉 = λx.Cv〈[t]〉

Figure 3.3: Call-by-name and call-by-value CPS transformations

The first step in transforming the function to continuation-passing style is to name and
sequentialize all the intermediate computations. We do not transform constant literals or other
primitives such as arithmetic operators. The only significant intermediate result in factorial
function is the recursive call to the function itself. We denote the result of the recursive call
with a fresh variable v:

fact n = if n ≡ 1
then 1
else let v = fact (n − 1) in n ∗ v

The next step is to introduce the continuation k, and redefine the original function as CPS-
transformed function applied to the identity continuation. The transformed program, after mov-
ing the lambda-abstraction parameter k to left-hand side of the helper function definition, looks
as follows:

factCPS :: Int → (Int → a)→ a
factCPS n k = if n ≡ 1

then k 1
else factCPS (n − 1) (λv→ k (n ∗ v))

fact :: Int → Int
fact n = factCPS n (λv→ v)

The resulting factorial function is tail-recursive. However, while the original program built a
high call stack, the transformed program builds a large closure.

In the derivation we have cheated a bit. Namely the host language is evaluated under call-
by-name strategy, but we have applied call-by-value CPS transformation. While this does not
not preserve the semantics of the original program in general case in the current case it’s correct
as the factorial function is strict.

25



3.2.2 Defunctionalization
Defunctionalization is a global code transformation that turns programs with higher-order func-
tions into first-order ones. We are using the defunctionalization as described by Reynolds
in [Rey98]. The method represents functions with data types, and replaces function abstrac-
tions with data constructors.

For every function space the transformation denotes the type of the function with a new
data type, and every lambda abstraction of that type becomes a data constructor in the new
data type. Free variables of lambda abstractions become parameters to the corresponding data
constructors. Function applications are represented with first-order functions, that take the
representation of the function and the function parameters as input, and perform the application
as dictated by the original lambda abstraction body. Correctness of defunctionalization has
previously been established by Nielsen in [Nie00].

Note that we do not consider the case of transforming polymorphic programs. We assume
that before defunctionalization is applied the program is converted into monomorphic form via
code duplication.

As an example we will apply defunctionalization to the higher-order tail-recursive factorial
function derived in the previous subsection. The example has only single function space of
type Int → Int that we will denote with a new data type Fun. Inhabitants of the only function
space are given rise to by two lambda abstractions: a) λv → v with no free variables; and
b) λv → k (n ∗ v) with n :: Int, and k :: Int → Int that will be represented with Fun data type,
as free variables. Thus we create a data type with two constructors, and an appropriately typed
application function:

data Fun = L0

| L1 Fun Int

app0 :: Fun→ Int → Int
app0 L0 v = v
app0 (L1 k n) v = app0 k (n ∗ v))

In the original program we have replaced the abstractions with data constructors and appli-
cations of higher-order functions with calls to the new first-order function. This gives us the
following defunctionalized program:

factDEF :: Int → Fun→ Int
factDEF n k = if n ≡ 1

then app0 k 1
else factDEF (n − 1) (L1 k n)

fact :: Int → Int
fact n = factDEF n L0

Note that the data structure has a list structure, and thus the new program can be made more
readable and intuitively understandable by using already existing list type:

app1 :: [Int ]→ Int → Int
app1 (n : k) v = app1 k (n ∗ v)
app1 [ ] v = v

factLIST :: Int → [Int ]→ Int
factLIST n k = if n ≡ 1

then app1 k 1
else factLIST (n − 1) (n : k)

fact :: Int → Int
fact n = factLIST n [ ]

By combining the continuation-passing style with defunctionalization we have reduced the
higher-order non-tail-recursive factorial function into two first-order tail-recursive functions.
One of which constructs a stack of values, and other multiplies the elements of the stack to-
gether. With further non-automatic optimizations it’s possible to reach a tail-recursive factorial
function which operates on integer accumulator, but those transformations are beyond the scope
of this work.

3.2.3 Thunk-based simulation

In Section 3.2.1 we saw that call-by-name evaluation strategy can be simulated in call-by-value
setting by performing a call-by-name CPS transformation. Turns out that there is a simpler
way to simulate CBN programs by transforming the program to explicitly suspend the compu-
tation of function arguments. We call a delayed or suspended computation (or parameterless
procedure) a thunk.

Formally the thunk-based simulation takes place in λ-calculus extended with constructs
to explicitly force or delay computations. The syntax for the extended lambda calculus is as
follows:

t ∈ Λτ

t ::= c | x | λx.t | t0 t1 | delay t | force t

The evaluation functions for extended language are obtained by adding following transition
rules to both the call-by-name, and the call-by-value operational semantics defined in Figure 3.1
and 3.2.

t 7−→ t′

force t 7−→ force t′
force (delay t) 7−→ t

The transformation to thunk-based form from regular λ-calculus is achieved by annotating
function parameters with explicit delay, and forcing the evaluation of variables.

T〈[·]〉 : Λ→ Λτ

T〈[c]〉 = c
T〈[x]〉 = force x

T〈[λx.t]〉 = λx.T〈[t]〉
T 〈[t0 t1]〉 = T〈[t0]〉(delay T〈[t1]〉)

An optimization to the transformation has been supplied by Danvy and Hatcliff in [DH92]. The
improved transformation is performed on λ-terms with strictness annotations. The parameters
to strict functions are not suspended, and thus the variables that have been introduced by strict
functions are not explicitly forced. This result allows for the thunk-based simulation to be
guided by the results of a strictness analysis.

Thunk-transformed closed terms are indifferent to evaluation strategy, and simulate call-by-
name evaluation in call-by-value setting:

evalv(T〈[t]〉) ' evaln(T〈[t]〉)
T〈[evaln(t)]〉 'τ evalv(T〈[t]〉)

Here the equivalence relation t1 'τ t2 means that the terms are both n-convertible, and v-
convertible given that the conversion rules are extended to account for the force and delay
primitives. The evaluation rules too need to account for the new primitives too.

Let us finally note that the call-by-name CPS transformation factors into composition of
thunk-based simulation followed by call-by-value CPS transformation. Informally: Cv ◦ T ≡

Cn, and the result was shown by Hatcliff and Danvy in [HD97].

3.3 The Hutton’s razor
Hutton’s razor [HW04] can be considered the simplest non-trivial expression language. The
language term is either integer literal, or addition of two terms. Abstract syntax of the lan-
guage is represented with Haskell data type, and the semantics of the language is defined by a
denotational evaluator. The evaluator directly corresponds to denotational semantics.

data Term
= Lit Int
| Add Term Term

eval :: Term→ Int
eval (Lit x) = x
eval (Add t0 t1) = eval t0 + eval t1

In order to demonstrate the methodology developed by Ager et al. in [ABDM03] we will,
as an introduction to it, derive an abstract machine from the given evaluator. First we note that
the eval function itself is not suitable as an abstract machine because it is not tail-recursive, and
is not presented as a term rewriting system.

3.3.1 CPS transformation
The first step is to perform continuation-passing style transformation to convert the evaluator
to tail-recursive form. We choose to evaluate the left-hand side of the addition operator first as
the semantics of the language has left the evaluation order of its parameters open.

eval0 :: Term→ (Int → a)→ a
eval0 (Lit x) k = k x
eval0 (Add t0 t1) k = eval0 t0 (λv0 → eval0 t1 (λv1 → k (v0 + v1)))
runEval0 :: Term→ Int
runEval0 t = eval0 t (λv→ v)

The equivalence of eval to runEval0 is due to the correctness of the CPS transformation.
While the evaluator is now linear, it’s no longer first-order. We have applied the call-by-value
CPS transformation, as the original program was strict. Defunctionalization is used next to
eliminate the use of higher-order functions.

28



3.3.2 Defunctionalization
The CPS transformed program has a single function space of type Int → Int. While the type of
evaluation function is more generic, it is specialized to integer at the call site in the runEval0

function. If the evaluation function was called on multiple different concrete types (for instance,
additionally eval0 t print) we would specialize the function for all types, and perform the
defunctionalization on all of the specializations individually.

The only function space shall be denoted with Fun data type. Within the single function
space we have three function abstractions:

1. λv→ v with no free variables;

2. λv1 → k (v0 +v1) with two free variables v0 ::Int, and k ::Int → Int that will be represented
with Fun; and

3. λv0 → eval0 t1 (λv1 → k (v0 + v1)) with two free variables t1 :: Term, and k :: Int → Int
that will be represented with Fun.

The enumeration of the lambda function constructors gives us the following data type to repre-
sent the function space:

data Fun
= L0

| L1 Int Fun
| L2 Term Fun

Every lambda function constructor within the evaluation function is replaced by the cor-
responding data constructor. Free variables of the lambda functions become arguments to the
data constructors.

eval1 :: Term→ Fun→ Int
eval1 (Lit x) k = app k x
eval1 (Add t0 t1) k = eval1 t0 (L2 t1 k)
runEval1 :: Term→ Int
runEval1 t = eval1 t L0

A new function app is used to apply the representation of the function space to actual values.
The body of the application function is derived from the definition of the lambda functions.
Note that the transformation has been applied to the lambda function bodies themselves, and
the inner lambda function has been eliminated this way:

app :: Fun→ Int → Int
app L0 v = v
app (L1 v0 k) v1 = app k (v0 + v1)
app (L2 t1 k) v0 = eval1 t1 (L1 v0 k)

Due to the correctness of the defunctionalization we have that runEval0 t ≡ runEval1 t
for every expression t. Technically the presented evaluator is in curried form, and thus is still
higher-order. However, because all of the function applications are total, we are not going to
uncurry the functions explicitly.

29



3.3.3 The abstract machine
The final evaluator is ready to be presented in an abstract machine form. Syntactically the
expression language terms are denoted with t, expressible values with v, and the evaluation
contexts with k. The language literal values v̄ are marked with an over-bar to distinguish them
from expressible values. Expressible values v are integers. Evaluation context k has a stack
structure, and is either empty stack nil, or has expressible value or expression as the topmost
element.

v ∈ Z
t ::= v̄ | t0 + t1

k ::= nil | v :: k | t :: k

Note that the abstract machine compared to the Haskell evaluator does not consider undefined
values ⊥, and can have arbitrary integers as expressible values.

The transition system has an initialization rule, two transition rules, and a stopping rule.
The two evaluation transitions directly correspond to the definition of eval function, and the two
application transitions correspond to the app function. Notice that the evaluation of addition
operator is very similar to evaluation of function application in the CEK abstract machine.

t ⇒eval 〈t, nil〉
〈t0 + t1, k〉 ⇒eval 〈t0, t1 :: k〉

〈v̄, k〉 ⇒eval 〈k, v〉
〈t1 :: k, v0〉 ⇒app 〈t1, v0 :: k〉
〈v0 :: k, v1〉 ⇒app 〈k, v0 + v1〉

〈nil, v〉 ⇒app v

The first app transition corresponds to the case where left-hand side of the operator has been
evaluated to v0, and the right-hand side t1 has not. In that case the system transition into
evaluating the right-hand side, and stores the value of the left-hand side on the stack. The
second rule corresponds to case where both sides of the operator have been evaluated, and in
that case the rule performs the operation, and possibly continues reducing expressible values
on the stack.

For example, to evaluate expression (1̄ + 2̄) + (4̄ + 5̄) we would follow the transition rules
as follows:

(1̄ + 2̄) + (4̄ + 5̄) ⇒eval

〈(1̄ + 2̄) + (4̄ + 5̄), nil〉 ⇒eval

〈(1̄ + 2̄), (4̄ + 5̄) :: nil〉 ⇒eval

〈1̄, 2̄ :: (4̄ + 5̄) :: nil〉 ⇒eval

〈2̄ :: (4̄ + 5̄) :: nil, 1〉 ⇒app

〈2̄, 1 :: (4̄ + 5̄) :: nil〉 ⇒eval

〈1 :: (4̄ + 5̄) :: nil, 2〉 ⇒app

〈(4̄ + 5̄) :: nil, 3〉 ⇒app

〈4̄ + 5̄, 3 :: nil〉 ⇒eval

〈4̄, 5̄ :: 3 :: nil〉 ⇒eval

〈5̄ :: 3 :: nil, 4〉 ⇒app

〈5̄, 4 :: 3 :: nil〉 ⇒eval

〈4 :: 3 :: nil, 5〉 ⇒app

〈3 :: nil, 9〉 ⇒app

〈nil, 12〉 ⇒app 12

Notice that the transitions respect left-to-right evaluation order, and the parameters to the oper-
ator are passed in correct order. This machine would correctly evaluate expressions even in the
case of non-associative, and non-commutative underlying operator.

There is no formal proof of correctness for the presented abstract machine, but there is
clear equivalence to the last evaluator in the derivation sequence. We can see that the abstract
syntax corresponds directly to the Term data type, and the stack structure exactly to the Fun
data type. The evaluation and application rules correspond to eval1, and app functions. The
initial transition corresponds to runEval1 function. Due to the obvious correspondence to the
last evaluator and its equivalence to the initial one we can conclude that the presented abstract
machine is indeed correct in respect to the initial evaluator.

Chapter 4

Deriving abstract machine for a dataflow
language

In Chapter 3 we gave overview of the methodology developed by Ager et al. in [ABDM03] for
deriving abstract machines from denotational evaluators. The notion of abstract machines was
defined in the Section 3.1. In this chapter we will apply the methodology to derive an abstract
machine for the comonadic denotational dataflow language evaluator defined in Section 2.3.

In this chapter we will first recap the previously defined comonadic evaluator and apply a
series of code transformations to it in order to reach an equivalent abstract machine. The first
transformation is to remove the higher-order functions from the value domain of the language.
We achieve this via defunctionalization. Next we will inline various components to simplify
the evaluator. At this point we will diverge slightly from the methodology and convert the
evaluator to use thunks in order to simulate the lazy host-language semantics. To reach a tail-
recursive form we will apply call-by-value CPS transformation. Finally we will defunctionalize
the introduced continuations in order to reach a first-order form.

We have diverged only slightly from the methodology. It would have been possible to avoid
thunk-based simulation, and perform call-by-name CPS transformation directly, but we found
the thunk-based simulation approach to be simpler to understand and apply.

4.1 Initial evaluator

Let us recap the evaluator presented in Section 2.3 that stands as stepping stone for the deriva-
tion process. The language term is either a variable, a literal integer, a lambda abstraction,
a function application, or a followed-by construct. Only the followed-by construct is unique,
and rest of the evaluator is standard for lambda calculus. In Haskell the abstract syntax tree is
defined as follows:

data Term
= Lit Int
| Var String
| Lam String Term
| App Term Term
| Fby Term Term

The expressible value is either co-Kleisli arrow from a value to value, or an integer. The
data structure to represent values and the denotational evaluator itself are defined as follows:

data Value
= I Int
| F (LV Value→ Value)

eval :: Term→ LV Env→ Value
eval (Lit n) de = I n
eval (Var x) de = counit de ! x
eval (Lam x t) de = F (λdv→ eval t (repair x ‘cmap‘ czip dv de))
eval (App t0 t1) de = unF (eval t0 de) (cobind (eval t1) de)
eval (Fby t0 t1) de = fby (eval t0 de) (cobind (eval t1) de)

Throught the derivation the type of the value domain will change, and thus the type of the
environment as well. In particular all of the transformations would need to be applied to the
values stored in the environment too. For this reason we will assume that the initial environment
is empty:

type Env = [(String,Value)]
ε :: Env
ε = [ ]

Using this assumption only the type of the initial environment will change throughout the
derivation and the implementation remains as is. How to handle the case where the initial
environment contains values has previously been covered by Ager, Danvy, and Midtgaard
in [ADM05] for monadic evaluators.

We would like to point out various characteristics of the evaluator that need to be addressed
in the derivation process before the abstract machine form is reached. In particular the evalu-
ator: a) contains higher-order function in the value domain; b) passes higher-order function in
the form eval t1 to cobind; c) is evaluated lazily due to the host-language semantics; d) is not
tail-recursive; and e) is not presented as a term rewriting system.

4.2 From higher-order functions to closures
As a first step we eliminate the use of higher-order functions from the expressible value domain.
Instead the value domain will store closures in the form of triples consisting of a variable, a
term, and an environment.

The transformation is performed by defunctionalizing the function domain LV Value →
Value in the value data type. The only inhabitants of this function domain are constructed in
the lambda abstraction case, and are consumed in the lambda application case. The lambda
function constructor has three free variables x :: String, t :: Term, and de :: LV Env.

The defunctionalization is performed as usual, but the resulting data type with only one
constructor is inlined into the value data type. This results in the following data structure:

data Value0 = I0 Int
| C0 String Term (LV Env0)

type Env0 = [(String,Value0)]

Finally, we inline the constructed application function into the evaluator. As a result we
have replaced the right-hand side of lambda expression where function objects are constructed,
and modified the application case where the constructed functions are consumed. Integer values
need to be constructed with the appropriate data constructor. This gives us following code for
the evaluator with rest unmodified:

eval0 :: Term→ LV Env0 → Value0

eval0 (Lit n) de = I0 n
eval0 (Lam x t) de = C0 x t de
eval0 (App t0 t1) de =

case eval0 t0 de of
C0 x′ t′ de′ →

eval0 t′ (repair x′ ‘cmap‘ czip (cobind (eval0 t1) de) de′)
...

The runEval function needs to be updated to call the new evaluation function, and reflect
the new value type:

runEval0 :: Term→ Int → Value0

runEval0 t i = eval0 t (ε ↑ i)

Definition 1. We say that two non-empty lists are equivalent, if they are of the same length
and are point-wise equivalent. Two functions are equivalent if they map equivalent inputs
to equivalent outputs. The equivalence relation between values of type Value, and Value0 is
defined as follows:

⊥ � ⊥ I n � I0 n
f � λdv′ → eval0 t (repair x ‘cmap‘ czip dv′ de)

F f � C0 x t de
.

Proposition 2 (Full correctness). For every language term t :: Term, and i > 0 if evaluating
runEval t i yields a value then runEval0 t i yields an equivalent value.

Proof. Due to correctness of defunctionalization. �

4.3 Inlining the comonad
The second step of the transformation is to inline some of the comonad specific functions. We
extract the environment updating from the application case of the evaluator, and introduce it as
a helper function:

update0 :: String→ Term→ LV Env0 → LV Env0 → LV Env0

update0 x′ t1 de de′ =

repair x′ ‘cmap‘ czip (cobind (eval1 t1) de) de′

First we will inline the definitions of cobind, czip and repair into the helper function, and
after some simplification we are left with following code:

35



update1 :: String→ Term→ LV Env0 → LV Env0 → LV Env0

update1 x′ t1 de@(Cons e es) (Cons e′ es′) =

Cons (insert x′ (eval1 t1 de) e′) (update1 x′ t1 es es′)
update1 x′ t1 de de′ = One (insert x′ (eval1 t1 de) (counit de′))

Proof of the equivalence of the two functions is presented in Appendix A.1.
As both update functions are equivalent we can replace the occurrences of right-hand side

of update0 with left-hand side of update1:

eval1 :: Term→ LV Env0 → Value0

...
eval1 (App t0 t1) de =

case eval1 t0 de of
C0 x′ t′ de′ → eval1 t′ (update1 x′ t1 de de′)

...

By introducing update1 function we have gotten rid of all uses of functions repair and czip.
We observe that in case of followed-by term the evaluation function is strict on the envi-

ronment parameter. This is due to the fact that both fby, and cobind are strict on the second
parameter. The strictness allows us to pattern match on the second evaluator parameter, and
simplify using simple case analysis. For singleton environments:

eval1 (Fby t0 t1) (One e)
≡ { definition of eval1 }

fby (eval1 t0 (One e)) (cobind (eval1 t1) (One e))
≡ { definition of cobind }

fby (eval1 t0 (One e)) (One (eval1 t1 (One e)))
≡ { definition of fby }

eval1 t0 (One e)

and for larger environments:

eval1 (Fby t0 t1) (Cons e de)
≡ { definition of eval1 }

fby (eval1 t0 (Cons e de))
(cobind (eval1 t1) (Cons e de)

≡ { definition of cobind }
fby (eval1 t0 (Cons e de))

(Cons (eval1 t1 (Cons e de)) (cobind (eval1 t1) de))
≡ { definition of fby }

counit (cobind (eval1 t1) de))
≡ { comonad law }

eval1 t1 de

The simplification gives us the following followed-by construct cases:

...
eval1 (Fby t0 t1) (One e) = eval1 t0 (One e)
eval1 (Fby t0 t1) (Cons e de) = eval1 t1 de

The updated runEval function simply calls the new evaluation function:

runEval1 :: Term→ Int → Value0

runEval1 t i = eval1 t (ε ↑ i)

Proposition 3 (Full correctness). For every language term t :: Term, and i > 0 if runEval0 t i
yields some value, then it’s equal to runEval1 t i.

Proof. Due to the correctness of applied simplifications. �

4.4 Introduction of thunks
The state of the evaluator we have after inlining has two issues. Firstly the newly constructed
update1 function is mutually recursive to eval1. This can not be the case if we wish it to become
primitive construct in the abstract machine.

Second issue is that the correct semantics of the evaluator depends on the non-strict se-
mantics of Haskell. Namely, the results of eval1 function applications performed in the update
function are inserted into environment but are only forced when evaluating variables. We wish
for the abstract machine to be indifferent to the evaluation order. One option is to apply call-by-
name CPS transformation to gain correct evaluation order indifferent and tail-recursive evalua-
tor, but we found thunk-based simulation to be a better option.

We will make the laziness explicit by introducing force, and delay constructs into the evalu-
ator. The constructs are placed based on an ad-hoc strictness analysis. We are going to assume
that terms, and environment lists passed to the evaluation function are always defined, and thus
can be strictly evaluated. The update, and counit functions are strict.

Following the assumptions we notice that the only non-strict function in the evaluator is
the environment insertion function. The insert is lazy in the value parameter, and the lookup
returns potentially delayed value. The only values inserted into the environment are the results
of calls to the evaluation function. Because of that we are not implementing generic suspension
operation, but specialize it.

force :: Value1 → Value1

force (T1 t de) = eval2 t de

delay :: Term→ LV Env1 → Value1

delay t de = T1 t de

Notice that for every term t, and environment de if eval2 t de yields some value, then it’s equal
to force (delay t de).

The suspended computations will be stored directly in the value domain as thunks.

data Value1

= I1 Int
| C1 String Term (LV Env1)
| T1 Term (LV Env1)

type Env1 = [(String,Value1)]

There might also arise question if we could modify the environment to include only unevaluated
thunks instead of arbitrary expressible values. The answer is yes, but we choose not to as the
delay operation can be slightly optimized. For example, there is no reason to delay literal values
and closures.

We will note that delayed values of type Value0 can also be expressed as functions from the
unit type () to the domain itself () → Value0. If this approach was taken the environment has
to contain functions of this form. The introduced higher-order functions have to be eliminated
using defunctionalization which will eventually, after merging the value domain with created
data type, result in exactly the same evaluator. The indirect approach through higher-order
functions is more formal, but is much more verbose.

Mutual recursion and implicit non-strict behaviour is now eliminated by introducing explicit
delay into the update1 function giving us:

update2 :: String→ Term→ LV Env1 → LV Env1 → LV Env1

update2 x′ t1 de@(Cons e es) (Cons e′ es′) =

Cons (insert x′ (delay t1 de) e′) (update2 x′ t1 es es′)
update2 x′ t1 de de′ =

One (insert x′ (delay t1 de) (counit de′))

Values are forced to evaluate only when looked up from the environment, and function appli-
cation invokes the new update2 function:

eval2 :: Term→ LV Env1 → Value1

...
eval2 (Var x) de = force (counit de ! x)
eval2 (App t0 t1) de =

case eval2 t0 de of
C1 x′ t′ de′ → eval2 t′ (update2 x′ t1 de de′)

...

Notice that evaluation function can never give rise to thunks. The rest of the evaluator remains
unmodified apart from updated types and data constructors.

The only place where values from the environment are read is the variable lookup. Previ-
ously the values inserted into the environment were evaluated when the result of the variable
lookup was required. This behaviour was due to the call-by-name reduction strategy of the host
language. Now we delay the evaluation explicitly, and force the evaluation of values when they
are actually looked up from the environment. Under non-strict semantics that’s correct.

We have gotten rid of mutual recursion of eval1 and update1, but introduced it to eval2 and
force. However, this is not an issue as force is conceptually much simpler function.

To update the runEval function the data type is updated, and the new evaluation function is
called:

runEval2 :: Term→ Int → Value1

runEval2 t i = eval2 t (ε ↑ i)

Definition 2. We say that two environments are equivalent if they are of equal length, and point-
wise map the same variable to equivalent values. The equivalence relation between values of

type Value0, and Value1 is defined as follows:

⊥ � ⊥ I1 n � I1 n

de � de′

C0 x t de � C1 x t de′

v1 � eval2 t de
v1 � T1 t de

Proposition 4 (Full correctness). For every language term t :: Term, and i > 0 if runEval1 t i
yields some value, then it’s equivalent to runEval2 t i.

Proof. Due to correctness of thunk-based simulation. �

Proposition 5 (Indifference). The evaluation function eval2 is indifferent to the host language
evaluation strategy.

Proof. Due to indifference property of the thunk-based simulation. �

4.5 CPS transformation
In this section we will apply call-by-value CPS transformation to the call-by-name thunk-
transformed evaluator. The previous and current transformation combined are equivalent to
call-by-name CPS transformation.

We have decided to convert Value1 to closure (Value1 → a) → a. The transformation is
straightforward and result in the following higher-order tail-recursive evaluator:

eval3 :: Term→ LV Env1 → (Value1 → a)→ a
eval3 (Lit n) de k = k (I1 n)
eval3 (Var x) de k = forceCPS (counit de ! x) k
eval3 (Lam x e) de k = k (C1 x e de)
eval3 (App t0 t1) de k = eval3 t0 de (λy0 →

case y0 of C1 x′ e′ de′ →
eval3 e′ (update x′ t1 de de′) k)

eval3 (Fby t0 t1) (One e) k = eval3 t0 (One e) k
eval3 (Fby t0 t1) (Cons e de) k = eval3 t1 de k

Due to mutual recursion the CPS conversion is also applied to the force function:

forceCPS :: Value1 → (Value1 → a)→ a
forceCPS (T1 t de) k = eval3 t de k

To execute the new evaluator we apply it to identity function:

runEval3 :: Term→ Int → Value1

runEval3 t i = eval3 t (ε ↑ i) (λv→ v)

Proposition 6 (Full correctness). For every language term t :: Term, and i > 0 if runEval2 t i
yields some value then runEval3 t i yields an equal value.

Proof. Due to the indifference property of eval2, and the correctness of call-by-value CPS
transformation. �

4.6 Defunctionalization
We have performed multiple steps of transformations all of which are covered in Subsec-
tion 3.2.2.

To remove the higher-order functions introduced by the CPS transformation we need to
defunctionalize the created closures. Turns out that the defunctionalization step is quite simple,
as the evaluator only has two lambda abstractions of the same type. One of them is the identity
function with no free variables, and the other has a term, an environment, and the closure itself
as free variables. The enumeration of abstractions gives us a data type with a list structure. The
elements of the closure representing list are pairs composed of a term, and an environment. We
are somewhat more general, and represent the closure with a list of values, while in reality the
only values will be thunks. The previous reasoning gives up the following type for the new
evaluation function:

eval4 :: Term→ LV Env1 → [Value1 ]→ Value1

Environment, and the value domain remain as is.
In the evaluator the continuation is applied in two positions. In the variable literal, and in

the lambda abstraction case. The evaluation function with transformed higher-order function
applications looks as follows:

eval4 (Lit n) de k = app k (I1 n)
eval4 (Var x) de k = forceDEF (counit de ! x) k
eval4 (Lam x e) de k = app k (C1 x e de)
eval4 (App t0 t1) de k = eval4 t0 de (T1 t1 de : k)
eval4 (Fby t0 t1) (One e) k = eval4 t0 (One e) k
eval4 (Fby t0 t1) (Cons e de) k = eval4 t1 de k

Defunctionalization of the forceCPS function simply transforms the type, and the code remains
as is:

forceDEF :: Value1 → [Value1 ]→ Value1

forceDEF (T t de) k = eval4 t de k

The application function for the only function space is constructed from lambda abstraction
bodies, and is defined as follows:

app :: [Value1 ]→ Value1 → Value1

app (T t1 de : k) (C x′ t′ de′) = eval4 t′ (update x′ t1 de de′) k
app [ ] v = v

Finally, the identity function in runEval3 needs to be transformed to empty list giving us:

runEval4 :: Term→ Int → Value1

runEval4 t i = eval4 t (ε ↑ i) [ ]

Proposition 7 (Full correctness). For every language term t :: Term and i > 0 if the expression
runEval3 t i yields some value v then it’s equal to runEval4 t i.

Proof. Due to correctness of defunctionalization. �

4.7 The final abstract machine

In this section we present the abstract machine in formal manner, and show that it’s equivalent
to the evaluator presented in previous section. The host language is λ-calculus extended with
the followed-by construct. A language term t is either variable x, literal n, lambda abstraction,
function application, or a followed-by construct.

t ::= x | n | λx.t | t0 t1 | t0 ← t1

The abstract machine operates on a non-empty stack of environments de, and an environ-
ment e maps variables to thunks. Thunks are pairs composed of a term, and an environment
stack.

de ::= e | e C de
e ::= ε | e[x 7→ T t de]

Note that a while environment itself may be empty, a sequence of environments is defined in
such manner that it’s not possible to construct an empty sequence.

The environment updating notion is extended to environment sequences. If the sequence is
a singleton then the only environment is updated as is. To update a non-singleton sequence of
environments with a thunk we need to pattern match on the environment sequence of the thunk.
If the thunk is stored with a history, then we update the first environment to map to given thunk,
but rest of the sequence is updated to map the variable to a same thunk but with shorter history.
If we map a variable to a thunk with singleton environment in a non-singleton environment
sequence, then the rest of the sequence is simply discarded resulting in singleton environment.

(e′ C de′)[x 7→ T t (e C de)] = e′[x 7→ T t (e C de)] C de′[x 7→ T t de]
(e′ C de′)[x 7→ T t e] = e′[x 7→ T t e]

The ↑ operator is used to map an environment to a specified sized stack of environments,
and counit is used to extract the first environment from a sequence.

e ↑ i ≡ e C . . .C e︸       ︷︷       ︸
i

, if i > 0

counit e ≡ e
counit (e C de) ≡ e

Variable lookup on non-singleton environment looks the variable up from the first environ-
ment:

de ! x ≡ counit de ! x .

Expressible value v is either an integer, a thunk, or a closure. Continuation stack k is
composed of values.

v ::= n | T t de | C x t de
k ::= nil | v :: k

Finally, the abstract machine has initialization rule for every positive integer i, final transi-
tion rule, and eight reduction rules. Evaluation rules operate on triples composed of a term, an
environment sequence, and a continuation stack. Application rules operate on pairs composed

of a continuation stack, and an expressible value.

t ⇒i
eval 〈t, ε ↑ i, nil〉 if i > 0

〈x, de, k〉 ⇒eval 〈t′, de′, k〉 where T t′ de′ = de ! x
〈n, de, k〉 ⇒eval 〈k, n〉

〈λx.t, de, k〉 ⇒eval 〈k, C x t de〉
〈t0 t1, de, k〉 ⇒eval 〈t0, de, T t1 de :: k〉
〈t0 ← t1, e, k〉 ⇒eval 〈t0, e, k〉

〈t0 ← t1, e C de, k〉 ⇒eval 〈t1, de, k〉
〈T t1 de :: k, C x′ t′ de′〉 ⇒app 〈t′, de′[x′ 7→ T t1 de], k〉

〈nil, v〉 ⇒app v

The abstract machine presented here corresponds directly to the evaluator defined in the
previous section. We note that: a) the abstract syntax corresponds to the Term data type;
b) expressible values corresponds to Value1 data type; c) the environment lookup corresponds
to counit de ! x expression; d) environment updating corresponds to the update function; and
e) the initialization transitions corresponds to the runEval4 function.

The abstract machine is very similar to Kirvne’s abstract machine apart from few differ-
ences. A minor difference is that our language contains regular variables, while the Kirvne’s
machine operated on terms with De Bruijn indices. Because of difference representation of vari-
ables our abstract machines expressible value domain contains closures. The second difference
is the two additional rules to handle the followed-by construct. Finally, the language operates
on non-empty sequence of environments, and uses overloaded notions of variable lookup, and
variable updating.

Due to the direct correspondence and Proposition 7 we can conclude that the machine pre-
sented in this section is an abstract machine for a higher-order call-by-name dataflow language.

Chapter 5

Conclusions

In this thesis we explore the behaviour of programming languages that manipulates primitive
values as infinite time-evolving streams of data. These programming languages are known as
dataflow languages, and are used to model electronic circuits and to control hardware devices
in real-time.

In particular we look at the operational behaviour of a simple higher-order call-by-name
dataflow language. The language is syntactically λ-calculus that is extended with a causal
dataflow language construct. The dynamic semantics of the language is given in comonadic
denotational form which, however, is unfit to convey the operational behaviour. One way to
impart the operational behaviour is either small-step or big-step operational semantics. As an
alternative to operational semantics we have chosen to present the language evaluator in a form
of an abstract machine.

An abstract machine is a term rewriting system that has few properties that make it easy
for a machine to evaluate. Rather informally a term rewriting system is an abstract machine
if it has clear initialization, and halting conditions; does not behave undeterministically; and
the rules are always applied to terms at their roots. In addition to that it must be efficient to
select which of the transition rules to apply. Abstract machines operate directly on the source
language terms. Our goal was to derive an abstract machine for a hihger-order comonadic
dataflow language.

Ager et al. developed a methodology in [ABDM03] for deriving abstract machines from
program language evaluators implemented in a denotational style. The methodology applies a
series of well understood code transformations to the evaluator until the sought form is attained.
Major steps of the transformation are the continuation-passing style (CPS) transformation to
convert the evaluator to a tail-recursive form, and the defunctionalization for eliminating the
use of higher-order functions. The defunctionalization transformation is applied twice: first
to convert the value domain of the language to contain closures instead of functions, and the
second time to eliminate the higher-order functions introduced by the continuation-passing
style transformation.

We apply the methodology to a comonadic language evaluator implemented in a non-strict
language Haskell. The denotational evaluator is by Uustalu and Vene from [UV05] and corre-
sponds directly to the formal denotational semantics of the language. Because we are working
in a non-strict host language we have to apply the call-by-name CPS transformation in the
derivation. However, we found that it’s simpler to split the call-by-name CPS transformation
into composition of thunk-based simulation and regular call-by-value CPS transformation. We

guide the thunk-based simulation using ad-hock strictness analysis to avoid inserting unneces-
sary evaluation delays.

The derivation presented in this work results in an abstract machine which is nearly equiva-
lent to the Krivine’s abstract machine for call-by-name λ-calculus. Only significant differences
are that the machine for the dataflow language has additional two rules for evaluating the causal
dataflow operator, and uses overloaded notion of environment lookup and updating. A minor
difference is that our language denotes variables by name while Krivine’s machines uses De
Bruijn indices. This tells us that operationally dataflow language functions exactly like call-
by-name lambda calculus, but represents the environment in more complicated manner to track
history of variables. We have also demonstrated that the methodology of deriving abstract
machines can be directly applied to an evaluator implemented in a non-strict language.

5.1 Future work
While the abstract machine does give insight into the operational behaviour of the language it
does not directly lead to an efficient implementation. Similar situation held with call-by-name
language evaluators as the Kirivne’s machine was well known before any efficient implemen-
tations emerged. Additionally, the fact that implementing functional reactive programming
in Haskell has proved to be very tricky suggests that the same holds for implementing call-
by-name functional dataflow languages. Recent results in developing efficient implementation
of functional reactive programming in Haskell [Jel11] suggests that these approaches can be
adopted.

The most significant time inefficiency of the dataflow language evaluator is due to the fact
that all of the values of the resulting stream are computed independent to the previous ones.
Intermediate results from the past computations could be reused. One, but not the most efficient,
approach would be memoisation.

The evaluator also suffers from memory inefficiency due to having to keep all previous
values of variables in memory. Either a type system, or program analysis could be used to
figure out how much of the history the evaluator needs to keep around. This would be major
efficiency gain as most dataflow programs only depend on finite number of previous values.
However, it’s very probable that this kind of analysis does not work well in the presence of
higher-order functions.

Abstraktne masin komonaadilisele
andmevookeelele

Magistritöö (30 EAP)

Jaak Randmets

Resümee

Klassikaline lähenemine programmeerimiskeeltele on töötada baastüüpidega (täisarvud, tõe-
väärtused) ühe väärtuse kaupa. Keel võib kohelda baasväärtustena arve, objekte või funkt-
sioone, aga keele operatsioonid käsitlevad neid harva hulgi. Väärtuste kollektsioonid ja neid
töötlevad operatsioonid on enamasti realiseeritud keele eraldi vahenditega, mis võimaldavad
konstrueerida baastüüpidest keerulisemaid tüüpe. Üheks eraldi klassiks on massiivitöötluse ja
andmeparalleelsed keeled, mille baasväärtusteks võivad olla jadad, maatriksid või isegi suvali-
sed n-mõõtmelised massiivid. Paljud nende keelte baasoperatsioonid töötlevad massiive punk-
tiviisiliselt. Teiseks eri klassiks on andmevookeeled, kus käsitletakse kõiki keele baasobjekte
kui lõputuid andmevooge. Seega on keele tüübid lõputud vood tüüpidest, literaalid tähistavad
konstantseid vooge väärtustest ja näiteks arimeetilised operatsioonid töötavad sisendvoogudel
punktiviisiliselt. Andmevookeeli kasutatakse mikroskeemide modelleerimiseks ja reaalajasüs-
teemide juhtimiseks. Käesolevas töös uurimegi selliseid andmevookeeli.

Me ütleme, et keel on kõrgemat järku, kui ta käsitleb funktsioone baasväärtustena mis tä-
hendab, et lubab kasutaja defineeritud funktsioonidel parameetriks võtta ja tagastada teisi funkt-
sioone. Funktsioon on kõrgemat järku, kui ta saab kas sisendiks või tagastab väärtusena funkt-
siooni. Andmevookeeltes on paraku kõrgemat järku funktsioonide esitamisega probleeme ja
paljud andmevookeeled kas ei käsitle neid esimest järku objektidena või nõuavad, et tüüpides
oleks ilmutatult öeldud kas parameetrina edastatud funktsioon käitub puhtalt punktiviisiliselt
või mitte.

Kõrgemat järku andmevookeeletes kirjutatud programmidele saab formaalse tähenduse an-
da kasutades kategooriateooriast tuntud mõisteid. Üks võimalus on seda teha kasutades nooli,
aga leidub ka konkreetsem viis komonaadide näol. Mõlemad semantikad on antud denotatsioo-
nilisel kujul ja edastavad seega hästi programmide tähendust, jättes aga kõrvale programmmide
arvutuskäigu.

Programmide operatsioonilise käitumise mõistmiseks on paar võimalust. Esimene neist
on anda keelele selle denotatsioonilise semantikaga samaväärne operatsiooniline loomulik või
struktuurne semantika. Teine võimalus on konstrueerida korrektne keele väärtustaja abstrakt-

se masina kujul. Käesoleva töö eesmärgiks on tuletada kõrgemat järku andmevookeelele selle
komonaadilise semantikaga samaväärne abstraktne masin.

Abstraktne masin on termide asendussüsteem, millel on mõned omadused, mis muudavad
selle väärtustamise arvutile lihtsamaks. Nimelt, termide asendussüsteemi nimetame abstrakt-
seks masinaks kui: sellel on selge alustamise reegel ning seiskamise reeglid, see käitub determi-
nistlikult, selle reeglite vasakud pooled ei sisalda korduvaid muutujaid ja reegleid rakendatakse
alati termide juurtel.

Abstraktse masina tuletamiseks kasutame metoodikat, mis seisneb algsele denotatsioonili-
sel kujul olevale väärtustajale järjest hästi tundutud kooditeisenduste rakendamises, kus abst-
raktse masina samaväärsus esialgse väärtustajaga tuletatakse rakendatud kooditeisenduste kor-
rektsusest. Suuremad sammud programmiteisenduste ahelas on: jätkuedastusstiili teisendus, et
jõuda sabarekursiivse väärtustajani, ning defunktsionaliseerimine, et teisendada programm esi-
mest järku kujule. Defunktsionaliseerimist rakendatakse sealjuures kaks korda: esimene kord,
et asendada interpretaatori väärtuste domeenis kõrgemat järku funktsioonid esimest järku su-
lundite esitusega ning teine kord, et eemaldada jätkuedastusstiili teisenduse poolt tekitatud kõr-
gemat järku funktsioonid.

Me rakendame kirjeldatud metoodikat komonaadilise semantikaga samaväärsele Haskellis
realiseeritud väärtustajale. Keele interpretaator on esitatud denotatsioonilises stiilis ning vastab
täpselt selle keele denotatsioonilisele semantikale. Lisaks peame arvestama ka sellega, et meie
interpretaator on realiseeritud mitteagaras keeles. Ainus kooditeisendus kus see oluliseks osu-
tub on jätkuedastusstiili teisendus, mis määrab kindlaks teisendatava programmi kontrollvoo.
Jätkuedastusstiili teisendusest on kaks versiooni: üks, mis säilitab nimekutse (call-by-name)
semantika ning teine, mis säilitab väärtuskutse (call-by-value) semantika.

Kuna meie väärtustaja on implementeeritud mitteagaras keeles, siis tuleb alloleva keele
nimekutse semantika säilitamiseks rakendada nimekutse jätkuedastusstiili teisendust. Lihtsam
lahendus on simuleerida nimekutse käitumist, teisendades programmi kujule, kus arvutusi vii-
vitatakse ning sunnitakse ilmutatult. Kuna programmid, kus kõik viivitused on esitatud ilmu-
tatud kujul, on sõltumatud väärtustamisjärjekorra suhtes, siis rakendame interpretaatorile väär-
tuskutse jätkuedastusstiili teisendust. Need kaks teisendussammu järjestikku rakendatuna on
samaväärsed nimekutse jätkuedastusstiili teisendusega.

Töös esitatud kooditeisenduste järjend annab meile esimest järku sabarekursiivse interpre-
taatori mitteagarale kõrgemat järku andmevookeelele. Esitame selle interpretaatori samaväärse
termide asendussüsteemina, millel on kõik abstraktse masina omadused. Tänu rakendatud koo-
diteisenduste korrektsusele ning viimase väärtustaja üksühesele vastavusele selle abstraktse
masinaga saame öelda, et toodud abstraktne masin on korrektne keele denotatsioonilise seman-
tika suhtes.

Esitatud abstraktne masin on väga sarnane hästi tuntud Krivine’i abstraktsele masinale.
Ainsateks oluliseks erinevusteks on kaks uut reeglit väärtustamaks andmevookeele spetsiifi-
list baasoperatsiooni ning muutujate keskkonna keerukam esitus. Kuna keskkonna esitus ei ole
sama, mis Krivine’i masinal, on seega ka keskkonnast otsimine ning selle uuendamine ülelaa-
ditud tähendusega. Tuletatud masina sarnasus hästi tuntud abstraktsele masinale annab tugeva
korrektsuse garantii ning ütleb, et mitteagarate andmevookeelte operatsiooniline käitumine on
sarnane tavaliste mitteagarate funktsionaalsete keelte käitumisele.

Appendix A

Extra proofs

A.1 Equivalence of update functions
Proposition 8. The following function:

update1 x′ t1 de@(Cons e es) (Cons e′ es′) =

Cons (insert x′ (eval1 t1 de) e′) (update1 x′ t1 es es′)
update1 x′ t1 de de′ = One (insert x′ (eval1 t1 de) (counit de′))

is equivalent to:

update0 x′ t1 de de′ =

repair x′ ‘cmap‘ czip (cobind (eval1 t1) de) de′

Proof. The proof is by induction over the length of argument lists. First we show that the
statement holds if the first argument is a singleton list and the second list of arbitrary length.
Let x′, t1, e, and de′ be some values of types directed by the type of the update0 function. Then
the following equalities hold:

update0 x′ t1 (One e) de′

≡ { definition of update0 }
repair x′ ‘cmap‘ czip (cobind (eval1 t1) (One e)) de′

≡ { definition of cobind }
repair x′ ‘cmap‘ czip (One (eval1 t1 (One e))) de′

≡ { definition of czip }
repair x′ ‘cmap‘ One (eval1 t1 (One e), counit de′)
≡ { definition of cmap and repair }

One (insert x′ (eval1 t1 (One e)) (counit de′))
≡ { definiton of update1 }

update1 x′ t1 (One e) de′

As for second case we show that the proposition holds if the first argument is of any length
and the second argument is a singleton list. Again, let the free variables be properly typed
arbitrary values:

update0 x′ t1 de (One e′)
≡ { definiton of update0 }

repair x′ ‘cmap‘ czip (cobind (eval1 t1) de) (One e′)
≡ { definiton of czip }

repair x′ ‘cmap‘ One (counit (cobind (eval1 t1) de), e′)
≡ { second comonad law }

repair x′ ‘cmap‘ One (eval1 t1 de, e′)
≡ { definition of repair and cmap }

One (insert x′ (eval1 t1 de) e′)
≡ { definition of counit }

One (insert x′ (eval1 t1 de) (counit (One e′)))
≡ { definiton of update1 }

update1 x′ de (One e′)

Let us assume that the statement holds for non-empty lists es and es′, and show that it also
holds for Cons e es, and Cons e′ es′ for any environemnt e, and e′. Let x′, and t1 be chosen
arbitrarily. The following reasoning holds:

update0 x′ t1 (Cons e es) (Cons e′ es′)
≡ { definition of update0 }

repair x′ ‘cmap‘ czip (cobind (eval1 t1) (Cons e es))
(Cons e′ es′)

≡ { definition of cobind }
repair x′ ‘cmap‘ czip (Cons (eval1 t1 (Cons e es)) (cobind (eval1 t1) es))

(Cons e′ es′)
≡ { definition of czip }

repair x′ ‘cmap‘ Cons (eval1 t1 (Cons e es), e′)
(czip (cobind (eval1 t1) es) es′)

≡ { definition of cmap }
Cons (insert x′ (eval1 t1 (Cons e es)) e′)

(repair x′ ‘cmap‘ czip (cobind (eval1 t1) es) es′)
≡ { induction assumption }

Cons (insert x′ (eval1 t1 (Cons e es)) e′) (update1 es es′)
≡ { definiton of update1 }

update1 x′ t1 (Cons e es) (Cons e′ es′)

It’s easy to see that the statement holds for undefined lists due to strictness of czip and
cmap. �

